```
(*working with the eigen energies of the 1D harmonic oscillator*)
```

In[30]:=

$$\text{explicitPartitionfunctionHO[beta\_]} := \sum_{n=0}^{\infty} \text{Exp}[-\text{beta}\ (n + 1/2)]$$

In[33]:=

```
FullSimplify[explicitPartitionfunctionHO[beta]]
```

Out[33]= $\frac{1}{2}\,\text{Csch}\left[\frac{\text{beta}}{2}\right]$

In[34]:=

$$\text{explicitEnergyHO[beta\_]} := -\partial_{\text{beta}}\,\text{Log[explicitPartitionfunctionHO[beta]]}$$

In[35]:=

```
FullSimplify[explicitEnergyHO[beta]]
```

Out[35]= $\frac{1}{2}\,\text{Coth}\left[\frac{\text{beta}}{2}\right]$

```
(*these are the 1D harmonic oscillator eigen states*)
```

In[46]:= $\text{EigenstatesHO[n\_, x\_, beta\_]} := \frac{1}{\sqrt{2^n\ (n!)\ \sqrt{\pi}}}\ \text{Exp}\left[-\frac{x^2}{2}\right]\ \text{HermiteH[n, x]}$

```
(*to get the density,
we need to square the eigen states and multiply by the wieght factor;
of course, we also need to normalize by the partition function*)
```

In[47]:=

$$\text{explicitDensityHO[x\_, beta\_]} := \frac{\sum_{n=0}^{\infty} (\text{EigenstatesHO[n, x, beta]})^2\ \text{Exp}[-\text{beta}\ (n + 1/2)]}{\text{FullSimplify[explicitPartitionfunctionHO[beta]]}}$$

```
(* we can try to see if Mathemtica can evaluate the infinite sum...
  turns out that Mathemtica does not know how to do this...*)
```

In[48]:=

```
FullSimplify[explicitDensityHO[x, beta]]
```

Out[48]= $2\,\text{Sinh}\left[\frac{\text{beta}}{2}\right] \sum_{n=0}^{\infty} \frac{2^{-n}\, e^{-\text{beta}\left(\frac{1}{2}+n\right)-x^2}\ \text{HermiteH}[n, x]^2}{\sqrt{\pi}\ n!}$

```
(*if we want to use the explicit expression,
we need to change the sum from infinity to a sufficiently large value;
here, I am using 100 (if the temperature is really high,
  then we need to include even more states in the sum)*)
```

In[49]:=

$$\texttt{explicitDensityCutoffHO[x\_, beta\_] :=} \frac{\sum_{n=0}^{100} (\texttt{EigenstatesHO[n, x, beta]})^2 \, \texttt{Exp[-beta (n + 1 / 2)]}}{\texttt{FullSimplify[explicitPartitionfunctionHO[beta]]}}$$

```
(*implementing the analytical expressions
 for the 1D harmonic oscillator from the notes*)
```

In[1]:=

$$\texttt{densityHO[x\_, beta\_] :=} \sqrt{\frac{\texttt{Tanh[beta / 2]}}{\pi}} \; \texttt{Exp}\left[-x^2 \, \texttt{Tanh[beta / 2]}\right]$$

In[18]:=

$$\texttt{partitionfunctionHO[beta\_] :=} \frac{\texttt{Exp[-beta / 2]}}{\texttt{1 - Exp[-beta]}}$$

In[17]:=

```
energyHO[beta_] := Coth[beta / 2] / 2
```

In[24]:=

```
Series[energyHO[beta], {beta, 0, 2}]
```

Out[24]= $\frac{1}{\text{beta}} + \frac{\text{beta}}{12} + O[\text{beta}]^3$

In[51]:=
```
Plot[{densityHO[x, 0.01], densityHO[x, 0.1],
  densityHO[x, 1], densityHO[x, 10]}, {x, -20, 20}]
```

Out[51]=



In[52]:=
```
Plot[{densityHO[x, 0.01], densityHO[x, 0.1]}, {x, -20, 20}]
```

Out[52]=



In[40]:=
```
Plot[{densityHO[x, 1], densityHO[x, 100]}, {x, -4, 4}]
```

Out[40]=



In[19]:=

```
Plot[energyHO[beta], {beta, 0, 10}]
```

Out[19]=



In[20]:=

```
Plot[energyHO[beta] /. beta → 1 / temp, {temp, 0, 1}]
```

Out[20]=



In[22]:=

```
Plot[partitionfunctionHO[beta], {beta, 0, 10}]
```

Out[22]=



In[54]:=

```
Plot[partitionfunctionHO[beta] /. beta → 1 / temp, {temp, 0.01, 1}]
```

Out[54]=




```
(*the following graph compares the analytical expression for
  the density from class with the "brute-force" implemention above
  (i.e., the implementation where I introduced a finite cutoff for the sum):
 as expected, the expressions give the same result!*)
```

In[55]:=

```
Plot[{explicitDensityCutoffHO[x, 0.9], densityHO[x, 0.9]}, {x, -5, 5}]
```

Out[55]=